PharmaSUG 2026 - Paper SI-227

A Statistical Programmer's Guide to Tipping Point Analysis in SAS

Ang Xu, Boehringer Ingelheim

ABSTRACT

Missing data due to treatment discontinuation poses challenges for longitudinal clinical trial analyses, as primary analyses often rely on unverifiable assumptions such as follow-the-reference (FTR). Sensitivity analyses are therefore necessary to assess robustness of conclusions to departures from these assumptions. This paper presents a practical implementation of tipping point analysis within a multiple imputation framework using SAS®. Non-monotone missing data are first transformed into a monotone structure, followed by reference-based multiple imputation and systematic delta adjustments to represent post-discontinuation deviations from FTR. Adjustment parameters are applied by treatment group and scaled by time since discontinuation. Each completed data set is analyzed using the primary model, with results combined using standard multiple-imputation inference. Results are summarized using tables and heatmaps to identify tipping points at which conclusions change, supporting transparent interpretation of sensitivity analyses.

INTRODUCTION

Missing data poses a persistent challenge in clinical trial analyses, particularly when treatment discontinuation occurs. Primary analyses often rely on assumptions that are inherently untestable, such as the follow-the-reference (FTR) assumption, where post-withdrawal outcomes are assumed to align with those observed in the reference group. While convenient, such assumptions may not fully reflect clinical reality, especially when treatment effects are expected to wane after discontinuation.

Sensitivity analyses are therefore essential to understand how robust primary conclusions are to these assumptions. One intuitive and transparent approach is the tipping point analysis, which systematically evaluates how large departures from the primary assumption must be in order to alter study conclusions.

From a statistical programming standpoint using SAS, key considerations include managing multiple layers of imputations, maintaining traceability across scenarios, intermediate data validation, and automating repetitive analyses. Emphasis is placed on reproducibility, efficiency, and clarity of output, making the approach scalable and adaptable to other studies and endpoints.

ANALYTICAL APPROACH

The tipping point analysis is conducted within a multiple imputation setting and follows three main stages as shown in Figure 1:



Figure 1. Analytical Approach

This structured approach with handling missing data, tipping point methodology, as well as analysis and inference allows systematic exploration of alternative post-withdrawal assumptions while maintaining consistency with the primary analysis model.

HANDLING MISSING DATA

Handling missing data consists of two steps:

- Transformation of non-monotone missing data into a monotone missingness pattern
- Multiple Imputation

Transformation of non-monotone missing data into a monotone missingness pattern

Outcome data (Table 1) sometimes exhibits non-monotone missingness, where patients may intermittently miss visits before returning for later assessments.

USUBJID	TRTP	AVISITN	AVISIT	AVAL	ABLFL
A00001	A	2	Baseline	87	Y
A00001	A	3	Week 4	86	
A00001	A	4	Week 8	93	
A00001	A	5	Week 12	92	
A00002	A	2	Baseline	78	Y
A00002	A	3	Week 4	82	

A00003	P	3	Week 4	86	
A00003	P	4	Week 8	95	
A00003	P	5	Week 12	82	
A00004	P	2	Baseline	76	Y
A00004	P	3	Week 4	83	
A00004	P	5	Week 12	81	

Table 1. Sample BDS structure ADaM Data

Because sequential regression imputation requires a monotone structure, an initial imputation step is performed to create data sets in which missingness, once initiated, remains missing for all subsequent time points.

```
proc transpose data=indata1a out=t_indata1a prefix=VAL;
   by USUBJID;
      var AVAL;
      id avisitn;
run;
```

Program 1 Code transposing the data to reveal the non-monotone missing pattern

After the transposing, the data would present one record per subject (Table 2).

USUBJID	VAL2	VAL3	VAL4	VAL5
A00001	87	86	93	92
A00002	78	82		
A00003		86	95	82
A00004	76	83		81

Table 2. Sample Data with non-monotone missing pattern

Please note that there are 3 patterns of missingness in the example:

- A00001 is a subject with all AVISIT complete
- A00002 is a subject with discontinuation
- A00003 is a subject with missing Baseline
- A00004 is a subject with particular AVISIT missing

At the next step, the MCMC method is used to transform the data set into a monotone missingness pattern by imputing intermediate missing values

```
PROC MI data=t_indata2a seed = 12345 nimpute = &N_IMPUTE. out = monotone noprint;
   VAR VAL2 COVAR VAL3 VAL4 VAL5;
      MCMC nbiter=1000 chain = multiple impute=monotone;
      BY TRTP;
RUN;

proc sort data=monotone; by _imputation_ TRTP USUBJID; run;
```

Program 2 Sample SAS code illustrating MCMC-based imputation to convert non-monotone missing data into a monotone structure prior to sequential regression imputation.

Each subject will be imputed for &N_IMPUTE times specified by protocol, the data is now presenting a monotone missing pattern (Table 3).

USUBJID	TRTP	_IMPUTATION_	VAL2	VAL3	VAL4	VAL5
A00001	A	1	87	86	93	92
A00001	A	2	87	86	93	92
A00001	A	3	87	86	93	92
A00002	A	1	78	82		
A00002	A	2	78	82		
A00002	A	3	78	82		
A00003	P	1	87.678	86	95	82
A00003	P	2	82.572	86	95	82
A00003	P	3	79.743	86	95	82
A00004	P	1	76	83	93.736	81
A00004	P	2	76	83	94.275	81
A00004	P	3	76	83	92.825	81

Table 3. Sample Data presenting monotone missing pattern

Please note that the intermediate missing value will be imputed by the PROC MI procedure above (A00003 and A00004).

Multiple Imputation

Multiple imputation is employed at this stage to preserve uncertainty due to missing data and to ensure stable inference. Reproducibility is emphasized through controlled imputation settings and consistent data handling. Data in each of the multiple-imputation generated data sets, now exhibiting a monotone missingness pattern, will be imputed once by using sequential regression.

```
proc mi data = monotone seed = 12345 nimpute = 1 out = outmi noprint;
   class TRTP COVAR;
      monotone reg;
      MNAR model(VAL3 VAL4 VAL5/ modelobs=(TRTP='Placebo'));
      VAR VAL2 COVAR VAL3 VAL4 VAL5;
      by _imputation_;
run;
```

Program 3 Sample SAS code illustrating monotone regression imputation with a reference-based MNAR model, using placebo arm data to impute post-baseline missing values.

After multiple imputation, the data set contains one record per subject within each imputation and missing values will be imputed referencing data from placebo arm (Table 4).

USUBJID	TRTP	_IMPUTATION_	VAL2	VAL3	VAL4	VAL5
A00001	A	1	87	86	93	92
A00001	A	2	87	86	93	92
A00001	A	3	87	86	93	92
A00002	A	1	78	82	94.125	90.576
A00002	A	2	78	82	94.135	93.875
A00002	A	3	78	82	93.987	88.257
A00003	P	1	87.678	86	95	82
A00003	P	2	82.572	86	95	82
A00003	P	3	79.743	86	95	82
A00004	P	1	76	83	93.736	81
A00004	P	2	76	83	94.275	81
A00004	P	3	76	83	92.825	81

Table 4. Sample data after multiple imputations

TIPPING POINT METHODOLOGY

Once monotone data sets are available, tipping point analysis is performed using a delta adjustment framework.

To represent departures from this assumption, a structured adjustment is added to the imputed values. A delta adjustment will be added to each imputed value. The value of each delta adjustment will be given by:

$$\delta_{ij} = S_i * \varphi_j$$

δ_{ij} = adjustment for a patient in treatment i at visit j,

i=1,2, where i =1 denotes the index for treatment arm, i=2 denotes the index for placebo arms,

j = VISIT 2,3,4,5

S_i= shift parameter for treatment i (Table 5)

φ_j= number of weeks between visit j-1 and visit j

δ_{ij} represents the assumed post-discontinuation deterioration for treatment group i at visit j. By scaling the adjustment with φ_j (time since previous assessment), the model reflects the clinically plausible assumption that longer time off treatment results in larger loss of effect.

Separate adjustment parameters are applied by treatment group, allowing flexible exploration of different post-withdrawal scenarios.

Shift Parameter (Example)	S1 (Treatment)	S2 (Placebo)
	-20	-20
	-10	-10
	0	0
	10	10
	20	20

Table 5. Sample of Shift Parameters Si per Treatment Group

These adjustments are designed to increase over time since treatment discontinuation, reflecting potential loss of treatment effect. The imputation process is sequential, meaning that adjustments accumulate when multiple visits are missing.

```
data outmi1;
   set outmi;
      if int(VAL3)^=VAL3 and TRTP="Treatment" and mVAL3="" then VAL3=VAL3+2*&sj.;
      else if int(VAL3)^=VAL3 and TRTP ="Placebo" and mVAL3="" then VAL3=VAL3+2*&sj2.;

      if int(VAL4)^=VAL4 and TRTP ="Treatment" and mVAL4="" then VAL4=VAL4+2*&sj.;
      else if int(VAL4)^=VAL4 and TRTP="Placebo" and mVAL4="" then VAL4=VAL4+2*&sj2.;

      if int(VAL5)^=VAL5 and TRTP="Treatment" and mVAL5="" then VAL5=VAL5+4*&sj.;
      else if int(VAL5)^=VAL5 and TRTP="Placebo" and mVAL5="" then VAL5=VAL5+4*&sj2.;

run;
```

Program 4 Sample SAS code implementing treatment- and visit-specific delta adjustments to imputed values for tipping-point sensitivity analysis.

ANALYSIS AND INFERENCE

Model Fitting and Estimation

At this step, each completed data set is analyzed using the same model as the primary analysis to ensure comparability. Treatment effects at the endpoint of interest are estimated, and results are combined across imputations using standard multiple imputation combining rules.

```
proc sort data=misp&grp.; by impid TRTP USUBJID avisitn visitnum; run;

proc MIXED data=misp&grp. cl method=reml order= internal;
   CLASS USUBJID TRTP COVAR(ref="0");
      MODEL AVAL= BASE COVAR  BASE*TIME_N TRTP  TRTP*TIME_N /solution CL ddfm=KR
singular=1e-10 outpred=pred_&grp. ; * TIME_N= DAY/1 year;
      RANDOM intercept TIME_N/ type=un subject=USUBJID G Gcorr V Vcorr solution;
      ESTIMATE 'Treatment - Placebo (1year)' TRTP*TIME_N 1 -1 / e CL;
      BY impid;

      ods output Estimates=est_imp_&grp.;

RUN;
```

Program 5 Sample SAS code fitting the primary mixed-effects model across imputed data sets for multiple-imputation inference.

The statistical result would be as Table 6

IMPID	LABEL	ESTIMATE	STDERR	PROBT	LOWER	UPPER
1	'Treatment – Placebo (1year)	-1.10	3.57	0.756	-10.02	4.56
2	'Treatment – Placebo (1year)	-3.45	4.12	0.689	-8.76	8.96
3	'Treatment – Placebo (1year)	-5.64	3.06	0.835	-9.08	9.63

Table 6 Treatment effect estimates across different imputations

And imputation-level estimates are pooled using standard multiple-imputation inference implemented in PROC MIANALYZE.

```
proc mianalyze data = est_imp_&grp.  ;
   by Label;
      modeleffects estimate;
      stderr stderr;
      ods output ParameterEstimates=par_imp&grp.;
run;
```

Program 6 Sample SAS code combining imputation-specific treatment effect estimates using standard multiple-imputation rules.

LABEL	NIMPUTE	ESTIMATE	STDERR	PROBT	LOWER	UPPER
'Treatment – Placebo (1year)	3	-3.80	3.89	0.864	-9.43	7.89

Table 7 Pooled treatment effect estimates

The pooled treatment effect shown corresponds to a specific pair of shift parameters, S1 and S2, from Table 5, the final output data would consist of a combination of n(S1)* n(S2) number of outcomes.

Interpretation of Results

The robustness of the primary findings is assessed by examining how extreme the departures from FTR must be to reverse conclusions. Scenarios requiring implausibly large adjustments suggest strong robustness.

Results are summarized in a structured table as well as a heatmap spanning all adjustment scenarios, allowing clear identification of tipping points at which statistical conclusions change.



Figure 2 Illustrative Tipping-point Heatmap

Moving below/right of the diagonal favors Treatment relative to Placebo; moving above/left favors Placebo.

This framework supports transparent communication of uncertainty to clinical, regulatory, and statistical stakeholders.

CONCLUSION

Tipping point analysis provides a structured and interpretable framework for sensitivity assessment under missing data. When implemented through multiple imputations, it allows systematic evaluation of unverifiable assumptions while preserving alignment with the primary analysis. This work demonstrates a practical programming workflow that supports robust and transparent inference in longitudinal clinical trials.

REFERENCES

International Council for Harmonisation. 2019. “ICH E9(R1): Addendum on Estimands and Sensitivity Analysis in Clinical Trials.” International Council for Harmonisation. Available at https://database.ich.org/sites/default/files/E9-R1_Step4_Guideline_2019_1203.pdf.

National Research Council. 2010. The Prevention and Treatment of Missing Data in Clinical Trials. Washington, DC: National Academies Press.

SAS Institute Inc. 2023. SAS/STAT® User’s Guide: The MI and MIANALYZE Procedures. Cary, NC: SAS Institute Inc.

CDISC. 2021. Analysis Data Model Implementation Guide (ADaM IG), Version 1.3. Austin, TX: Clinical Data Interchange Standards Consortium.

ACKNOWLEDGMENTS

The author would like to thank **Peitao Wu**, **Brian Vogler**, and **Hao Li** (Boehringer Ingelheim), **Trias Rakhmawati** (Syneos Health), **Sara Cameron** and **Matthias Post** (Mainanalytics GmbH) for their valuable contributions to this work. Their insightful discussions, methodological feedback, and constructive review greatly improved the clarity, rigor, and presentation of the analysis.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Ang Xu
Boehringer Ingelheim
ang_3.xu@boehringer-ingelheim.com